*****SUPPLEMENTARY INFORMATION**************
//Full ologit table
esttab /// 
	olo_fftax ///
	olo_fftax_Int_expressive ///
	olo_fftax_Int_neg_part ///
	olo_renew ///
	olo_renew_Int_expressive ///
	olo_renew_Int_neg_part ///
	olo_hhapp ///
	olo_hhapp_Int_expressive ///
	olo_hhapp_Int_neg_part ///
	using "tabs/pol_ologit.rtf", ///
	b(%3.2f)  se(%3.2f) ///
	star(* 0.05 ** 0.01) ///
	replace unstack nobase ///
	drop(cut*) ///
	stats(N r2o aic bic, fmt(0 2 1 1)) ///
	label
	
esttab /// 
	olo_offsets ///
	olo_offsets_Int_expressive ///
	olo_offsets_Int_neg_part ///
	olo_commute ///
	olo_commute_Int_expressive ///
	olo_commute_Int_neg_part ///
	olo_meat ///
	olo_meat_Int_expressive ///
	olo_meat_Int_neg_part ///
	olo_bath ///
	olo_bath_Int_expressive ///
	olo_bath_Int_neg_part ///
	using "tabs/behav_ologit.rtf", ///
	b(%3.2f)  se(%3.2f) ///
	star(* 0.05 ** 0.01) ///
	replace unstack nobase ///
	drop(cut*) ///
	stats(N r2o aic bic, fmt(0 2 1 1)) ///
	label
	
//Collinearity Diagnostics
local controls ///
	clim_con ///
	trust_soc trust_ins trust_sci ///
	female b6.age ///
	b1.racethn b3.region ///
	b2.educ income ///
	rural_urb	

loc iv ///
	party_id ///
	expressive ///
	neg_part
	
loc pol ///
	fftax ///
	renew ///
	hhapp
	
local behav ///
	offsets ///
	commute ///
	meat ///
	bath 

regress ///
	clim_pol_fftax ///
	`iv' ///
	`controls'

estadd vif
est sto vif_total

foreach x of local iv {
regress ///
	clim_pol_fftax ///
	`x' ///
	`controls'

estadd vif
est sto vif_`x'
}
esttab ///
	vif_total vif_party_id ///
	vif_expressive vif_neg_part ///
	using "tabs/vif.tex", ///
	cells(vif(fmt(2))) replace ///
	nomtitle nonumber l
	
//Interactions robustness checks
**Policy Support
foreach y of loc pol {
	qui sum clim_pol_`y', d
	loc `y'_max=r(max)
	qui ologit clim_pol_`y' ///
		c.party_id##c.(expressive ///
			c.neg_part `controls')
	est sto supp_`y'_Int_neg_part
	qui ologit clim_pol_`y' ///
		c.party_id##c.(expressive ///
			c.neg_part `controls')
	est sto supp_`y'_Int_expressive
	foreach x in expressive neg_part {
	est res supp_`y'_Int_`x'
	qui fre clim_pol_`y' if e(sample)
		loc valid_`y'=r(valid)[5,1]
		local tot_`y'=r(N)
		loc vperc_`y'=`valid_`y''/`tot_`y''
		di `vperc_`y''
	est res supp_`y'_Int_`x'
	qui margins, ///
		at(`x'=(-2(1)2) party_id=(1 7)) ///
		pr(out(``y'_max'))
	marginsplot, ///
			plot1opts(lc(maroon) m(th) mc(maroon)) ///
			plot2opts(lc(edkblue) m(dh) mc(edkblue)) ///
			ci1opts(lc(maroon%66)) ///
			ci2opts(lc(edkblue%66)) ///
			legend( ///
				order(3 "Republican" 4 "Democrat") ///
				pos(6) row(1)) ///
			yti("Pr('Strongly in favor')", s(small)) ///
			ylab(0(0.1)0.4, format(%3.2f) nogrid) ///
			yline(`vperc_`y'', lc(gs12)) ///
			xlab( ///
				-2 "-2Sd" -1 "-1Sd" 0 "Mean" 1 "+1Sd" 2 "+2Sd", nogrid) ///
			ti("") ///
			name(supp_`y'_`x'_int, replace)	nodraw	
	}
grc1leg ///
	supp_`y'_neg_part_int ///
	supp_`y'_expressive_int, ///	
	row(1) ///
	ycomm name(supp_`y'_int, replace) ///
	ti("`: var label clim_pol_`y''")
gr close supp_`y'_int
gr close supp_`y'_int
}

grc1leg ///
	supp_fftax_int ///
	supp_renew_int ///
	supp_hhapp_int, ///
	col(1) ysize(10) ///
	name(supp_pol_interaction, replace)
	
gr di, ysize(10) xsize(6.5)

gr export "figs/supp/pol_interaction_all.pdf", ///
	as(pdf) name("supp_pol_interaction") replace	
	
**Behaviors Interactions
foreach y of loc behav {
	qui sum `y', d
	loc `y'_max=r(max)
	qui ologit `y' ///
		c.party_id##c.(expressive ///
			c.neg_part `controls')
	qui r2o
	estadd scalar r2o=r(r2o)
	est sto supp_`y'_Int_neg_part
	qui ologit `y' ///
		c.party_id##c.(expressive ///
			c.neg_part `controls')
	qui r2o
	estadd scalar r2o=r(r2o)
	est sto supp_`y'_Int_expressive
	foreach x in expressive neg_part {
	est res supp_`y'_Int_`x'
	qui fre `y' if e(sample)
		loc valid_`y'=r(valid)[5,1]
		local tot_`y'=r(N)
		loc vperc_`y'=`valid_`y''/`tot_`y''
		di `vperc_`y''
	est res supp_`y'_Int_`x'
	qui margins, ///
		at(`x'=(-2(1)2) party_id=(1 7)) ///
		pr(out(``y'_max'))
	marginsplot, ///
			plot1opts(lc(maroon) m(th) mc(maroon)) ///
			plot2opts(lc(edkblue) m(dh) mc(edkblue)) ///
			ci1opts(lc(maroon%66)) ///
			ci2opts(lc(edkblue%66)) ///
			legend( ///
				order(3 "Republican" 4 "Democrat") ///
				pos(6) row(1)) ///
			yti("Pr('Very Willing')", s(small)) ///
			ylab(0(0.2)0.6, format(%3.2f) nogrid) ///
			yline(`vperc_`y'', lc(gs12)) ///
			xlab( ///
				-2 "-2Sd" -1 "-1Sd" 0 "Mean" 1 "+1Sd" 2 "+2Sd", nogrid) ///
			ti("") ///
			name(supp_`y'_`x'_int, replace)	nodraw	
	}
grc1leg ///
	supp_`y'_neg_part_int ///
	supp_`y'_expressive_int, ///	
	row(1) ///
	ycomm name(supp_`y'_int, replace) ///
	ti("`: var label `y''")
gr close supp_`y'_int
gr close supp_`y'_int
}

grc1leg ///
	supp_offsets_int ///
	supp_commute_int ///
	supp_meat_int ///
	supp_bath_int, ///
	col(1) ysize(10) ///
	name(supp_behav_interaction, replace)
	
gr di, ysize(10) xsize(6.5)

gr export "figs/supp/behav_interaction_all.pdf", ///
	as(pdf) name("supp_behav_interaction") replace
